
K A P I T O L A  7 

Vytváření interaktivních 
grafů

V této kapitole najdete:

Představení konceptu interaktivních grafů
Vytvoření automaticky se doplňujícího grafu – grafu, který se automaticky aktuali-
zuje, přidáte-li nebo odstraníte data
Použití posuvníku k určení dat řady
Použití rozevíracího seznamu k výběru počátečního a koncového bodu řady
Zobrazení posledních n bodů řady
Zobrazení každého n-tého bodu řady
Použití zaškrtávacích políček nebo rozevíracího seznamu k výběru zobrazené řady.

Termín interaktivní graf, jak jej používám v této knize, označuje graf, který se automatic-
ky mění v závislosti na prostředí listu. V jistém smyslu jsou interaktivní všechny grafy, 
protože jejich řady jsou propojené s oblastmi a grafy se automaticky aktualizují, změní-
te-li data v těchto oblastech. Ale o tento typ interaktivity v této kapitole nepůjde.

224 Část II: Ovládnutí grafů

Představení interaktivních grafů
Vytváříte-li tabulky, které používají i další lidé, pravděpodobně v této kapitole najdete několik 
užitečných metod. Mnoho těchto příkladů má jediný cíl: ulehčit uživatelům (především nováč-
kům) práci se sešity obsahujícími grafy.
Mezi interaktivní grafy patří například:

Graf, který se sám aktualizuje, aby použil data přidaná na konec řady.
Graf, který se sám aktualizuje, odstraníte-li nějaká data.
Graf, který omezuje množství zobrazených bodů řady (např. pouze posledních 12 bodů 
řady).
Graf, který zobrazí řadu v závislosti na hodnotě zadané do buňky nebo v závislosti na 
položce vybrané z rozevíracího seznamu.

Tato kapitola vám poskytne všechny potřebné informace k vytvoření několika typů grafů, které 
se automaticky aktualizují na základě informací obsažených v sešitu. Také se dozvíte, jak použí-
vat ovládací prvky dialogů (např. zaškrtávací políčka a rozevírací seznamy), abyste vytvořili 
interaktivní grafy. Žádný z příkladů v této kapitole nepoužívá makra.

Jiný způsob vytvoření interaktivního grafu představuje použití kontingenčního grafu. Další 
informace o kontingenčních grafech najdete v kapitole 11.

Z praxe: Vytvoření automaticky se doplňujícího 

grafu
Jeden z nejčastějších dotazů ohledně grafů zní: „Jak vytvořím graf, který se automaticky rozší-
ří, přidám-li do listu nová data?“
Abyste tento problém pochopili, podívejte se na obrázek 7.1 znázorňující list nastavený tak, aby 
uchovával denně aktualizované prodejní záznamy. Graf zobrazuje všechna data v listu. Zadáte-
li nová data, datová řada musí být ručně rozšířena, aby zahrnovala i nová data. Na druhou stra-
nu, pokud nějaká data odstraníte, datová řada grafu by se měla zúžit, aby se vynechaly odstra-
něné buňky.
Nebylo by příjemné, kdyby se datová řada rozšiřovala a zužovala automaticky?
Dobrou zprávou je, že tabulky v Excelu jsou si „grafu vědomé“. Tzn., že, pokud se data vašeho 
grafu nacházejí v tabulce (vytvořené pomocí Vložení  Tabulky  Tabulka), graf vždy zobrazí 
všechna data tabulky, a to i poté, co se tabulka rozšíří. Jinými slovy, máte-li data v tabulce, jsou 
pro vás informace v této části bezvýznamné.

Poznámka: Chystáte-li se svůj sešit sdílet s někým, kdo používá předchozí verzi Excelu, věci se mohou 
trochu zkomplikovat. Excel 2003 s tabulkou Excelu 2007 dokáže pracovat a program s ní zachází jako 
se seznamem (grafy používající data v seznamu Excelu 2003 se automaticky doplňují). Otevřete-li 
však soubor v Excelu nižší verze než 2003, graf se nebude automaticky doplňovat. Jinými slovy, pro 
vytvoření automaticky se doplňujícího grafu musíte použít postup popsaný v této části.

Kapitola 7: Vytváření interaktivních grafů 225

Obrázek 7.1: Kdyby byl graf automaticky se doplňující, automaticky by se aktualizoval, 
pokud byste zadali nová data.

Kapitola 3 popisuje několik způsobů úpravy zdrojových dat použitých v datové řadě grafu. 
Ačkoli žádná z těchto metod není obzvlášť složitá, každá z nich vyžaduje ruční zásah. Vytvoře-
ní automaticky se doplňujícího grafu vyžaduje o něco větší úsilí, ale výsledkem je graf, který se 
automaticky aktualizuje bez jakéhokoliv nutného zásahu uživatele.

Jednou z možností je, samozřejmě, určení větší než nutné oblasti datové řady. Příklad tohoto 
řešení, kdy datová řada zahrnuje i prázdné buňky, které budou v budoucnu zaplněny, je vidět 
na obrázku 7.2. Výsledkem je však nevyvážený graf obsahující spoustu volného prostoru. Ve 
většině situací by toto řešení neobstálo.

Obrázek 7.2: Určení prázdných buněk v oblasti dat obvykle není vhodné řešení.

226 Část II: Ovládnutí grafů

Jako úvod do světa interaktivních grafů obsahují následující části příklad z praxe, popsaný krok 
za krokem. Nejprve vytvoříte běžný graf a pak provedete úpravy nutné pro to, aby se graf auto-
maticky rozšířil, přidáte-li nová data, anebo se zúžil, pokud data odstraníte. 
Příklad využívá jednoduchého listu obsahujícího data dnů ve sloupci A a prodejní čísla ve sloup-
ci B. Předpokládá se, že nové datum a prodejní číslo přibude každý den a že graf zobrazí všech-
na data.

Vytvoření grafu

Prvním krokem je vytvoření standardního grafu používajícího data, která v současné době exis-
tují. Obrázek 7.1, představený výše, zobrazuje data a sloupcový graf, který z nich byl vytvořen.

Příklady v této části jsou dostupné na přiloženém CD. Jméno souboru je každodenní.xlsx.

Graf obsahuje jedinou datovou řadu, jejíž vzorec SADA vypadá následovně:
=SADA(List1!$B$1;List1!$A$2:$A$29;List1!$B$2:$B$9;1)

Tento vzorec SADA určuje, že:
Název datové řady je v buňce B1.
Popisky kategorie jsou v A2:A9.
Hodnoty jsou v B2:B9.

Zatím jde jen o obyčejný graf. Přidáte-li nové datum a hodnotu, v grafu se nová data nezo brazí. 
Ale to se brzy změní.

Vytvoření pojmenovaných vzorců 

V tomto kroku vytvoříte dva pojmenované vzorce. Názvy těchto vzorců budou posléze sloužit 
jako parametry vzorce SADA. V případě, že neznáte koncept pojmenovaných vzorců, je vysvět-
len později v této části. Chcete-li pojmenované vzorce, postupujte následovně:

1. Vyberte Vzorce  Definovat název  Definovat název, čímž otevřete dialogové okno Nový 
název.

2. Do pole Název zadejte Datum. Do pole Odkaz na zadejte tento vzorec:
=POSUN(List1!$A$2;0;0;POČET2(List1!$A:$A)-1;1)

3. Klepnutím na tlačítko OK vytvoříte vzorec s názvem Datum.
 Všimněte si, že funkce POSUN odkazuje na popisek první kategorie (buňka A2) a použí-

vá funkci POČET2 k určení počtu popisků v tomto sloupci. Jelikož má sloupec A záhlaví 
v prvním řádku, vzorec od čísla odečítá 1.

Nyní následujícím postupem vytvořte název pro prodejní data:
1. Vyberte Vzorce  Definovat název  Definovat název, čímž otevřete dialogové okno Nový 

název.
2. Do pole Název zadejte Prodej. Do pole Odkaz na zadejte tento vzorec:

=POSUN(List1!$B$2;0;0;POČET2(List1!$B:$B)-1;1)

Kapitola 7: Vytváření interaktivních grafů 227

 V tomto případě funkce POSUN odkazuje na první bod řady (buňka B2). Funkce 
POČET2 je zde použita k určení počtu bodů řady a je upravena s ohledem na popisek 
v buňce B1.

3. Klepnutím na tlačítko OK vytvoříte vzorec s názvem Prodej.
Poté, co dokončíte oba postupy, sešit bude obsahovat dva nové názvy, Datum a Prodej.

Úprava řady

Posledním krokem je úprava grafu tak, aby používal oba nové názvy namísto pevně daných 
odkazů na oblast. Postupujte následovně:

1. Aktivujte graf a vyberte Nástroje grafu  Návrh  Data  Vybrat data, čímž otevřete dia-
logové okno Vybrat zdroj dat (viz obrázek 7.3).

Obrázek 7.3: Pomocí dialogového okna Vybrat zdroj dat můžete změnit 
odkazy použité v datových řadách.

2. Označte v seznamu Položky legendy (řady) řadu Prodej a klepněte na tlačítko Upravit, 
čímž otevřete dialogové okno Upravit řady.

3. Do pole Hodnoty řad zadejte List1!Prodej.
4. Klepnutím na tlačítko OK zavřete dialog Upravit řady a vraťte se do dialogu Vybrat 

zdroj dat.
5. Klepněte na tlačítko Upravit pro Popisky vodorovné osy (kategorie), čímž otevřete dia-

logové okno Popisky osy.
6. Do pole Oblast popisku osy zadejte List1!Datum.
7. Klepnutím na tlačítko OK se vraťte do dialogu Vybrat zdroj dat.
8. Klepnutím na tlačítko OK zavřete dialog Vybrat zdroj dat.

Všimněte si, že v krocích 3 a 6 názvu předcházel název listu spolu s vykřičníkem. Jelikož pojme-
nované vzorce standardně fungují jako globální názvy sešitu (ne jako místní názvy listu), měli 
byste (technicky vzato) zadat název sešitu, vykřičník a poté samotný definovaný název. Excel 
je v tomto ohledu ale velmi shovívavý a opraví to za vás. Podíváte-li se do vzorce SADA (nebo 

228 Část II: Ovládnutí grafů

znovu otevřete dialog Vybrat zdroj dat), zjistíte, že Excel odkazy na list, které jste zadali, nahra-
dil názvem sešitu:
=každodenní.xlsx!Prodej

Zkrátka – používáte-li tyto pojmenované vzorce, můžete před jejich název vložit buď název 
listu, anebo název sešitu. (Mně připadá jednodušší použít název listu). Mějte však na paměti, 
že, pokud název listu nebo sešitu obsahuje znak mezery, musíte jej uzavřít do jednoduchých 
uvozovek, např. takto:
=‘každodenní prodej.xlsx‘!Prodej

nebo 
=‘prodejní záznamy‘!Prodej

Další informace o definovaných názvech najdete v následujícím rámečku „Jak Excel zachází 
s definovanými názvy“.

Tip: Alternativou k použití dialogu Vybrat zdroj dat je přímá úprava vzorce SADA grafu. Upravený 
vzorec SADA vypadá následovně:

=SADA(List1!$B$1;každodenní.xlsx!Datum;každodenní.xlsx!Prodej;1)

Jak Excel zachází s defi novanými názvy

Excel podporuje dva typy definovaných názvů: globální název sešitu a místní název listu. Oborem glo-
bálního názvu sešitu je celý sešit. Obyčejně, když vytvoříte definovaný název buňky či oblasti, můžete 
tento název použít na libovolném listu.

Můžete však vytvořit místní název listu. Součástí místního názvu listu je název listu. Příkladem může být 
List1!Data. Vytvoříte-li tento název, můžete jej použít ve vzorcích na Listu1 bez označení listu. Např.:

=Data*4

Pokud však tento název zadáte na jiném listu, Excel ho nerozpozná, dokud jej neupřesníte:

=List1!Data*4

Místní názvy listu jsou užitečné, neboť vám umožňují použít stejný název v odlišných listech. Můžete 
tak vytvořit místní názvy listu jako List1!Úrok, List2!Úrok a List3!Úrok. Každý název odpovídá buňce na 
svém vlastním listu. Vzorec používající název Úrok tedy použije formulaci pro svůj vlastní list.

Pojmenované vzorce v této kapitole jsou globální názvy sešitu, protože jim nepředchází název listu. 
Zadáte-li však jejich název do pole v dialogu Vybrat zdroj dat, Excel (z nějakého důvodu) vyžaduje, 
abyste název upřesnili buď pomocí názvu listu, nebo sešitu.

Test automaticky se doplňujícího grafu

Chcete-li si otestovat výsledky svého úsilí, zadejte do sloupců A a B nová data nebo odstraňte 
data z konce těchto sloupců. Pokud jste předchozí postupy provedli správně, graf se automatic-
ky aktualizuje. Obdržíte-li chybové hlášení nebo se graf neaktualizuje, pečlivě předchozí kroky 
překontrolujte.

Upozornění: Upravíte-li oblast použitou grafem tažením zvýrazňovacího rámečku, definované 
názvy (Datum a Prodej) budou nahrazeny skutečnými odkazy na buňky a graf se přestane auto-
maticky rozšiřovat a zužovat.

Jak automaticky se doplňující grafy fungují

Mnoho lidí používá metodu automaticky se doplňujících grafů, aniž by plně chápali, jak fungu-
je. Na tom samozřejmě není nic špatného. Pokud jste si prošli praktické cvičení uvedené výše, 
měli byste být schopni uplatnit postup i ve svých vlastních grafech. Pokud však pochopíte, jak 
to funguje, budete moci zajít dál, za rámec základního konceptu a vytvořit ještě schopnější typy 
dynamických grafů.

O pojmenovaných vzorcích

Mnoho metod interaktivních grafů popsaných v této kapitole využívá mocný nástroj zvaný 
pojmenované vzorce. Koncept definovaných názvů buněk a oblastí je vám asi znám, ale věděli 
jste, že definice názvů buněk a oblastí je ve skutečnosti špatné označení? Vytvoříte-li totiž název 
oblasti, ve skutečnosti vytvoříte pojmenovaný vzorec.
Pracujete-li s dialogovým oknem Nový název, pole Odkaz na obsahuje vzorec a pole Název 
obsahuje definovaný název vzorce. Časem zjistíte, že pole Odkaz na začíná vždy znakem rovná 
se, což je jasný důkaz, že jde o vzorec.
Na rozdíl od normálního vzorce, pojmenovaný vzorec neexistuje v buňce. Místo toto existuje 
v paměti Excelu a nemá ani adresu buňky. Výsledek pojmenovaného vzorce však můžete získat 
zadáním jeho názvu buď do standardního vzorce, nebo do vzorce SADA grafu. 
Poté, co definujete zmíněné dva pojmenované vzorce, Excel tyto vzorce vyhodnotí pokaždé, 
když je list vypočítán. Jelikož tyto pojmenované vzorce nejsou použité v žádné buňce, nemá 
jejich vytvoření za následek žádný viditelný efekt – alespoň do té doby, než je použijete k defi-
nování datové řady grafu.
Abyste se s pojmenovanými vzorci naučili lépe zacházet, použijte dialog Nový název k vytvoře-
ní následujícího vzorce, který pojmenujte Součet12buněk:
=SUM($A$1:$A$12)

Poté, co tento pojmenovaný vzorec vytvoříte, zadejte do libovolné buňky následující vzorec:
=Součet12buněk

Tento vzorec vrátí součet A1:A12.

O funkci POSUN

Klíčem k ovládnutí samo-rozvíjejících grafů je pochopení funkce POSUN. Tato funkce vrátí 
oblast, která je „posunuta“ od určené referenční buňky. Parametry funkce POSUN vám umož-
ňují určit vzdálenost od referenční buňky a rozměry oblasti (počet řádků a sloupců).
Funkce POSUN má následujících pět parametrů:

Odkaz: První parametr funkce POSUN je referenční buňka, vůči které provádíte 
posun. 
Řádky: Tento parametr určuje, o kolik řádků se má výsledná oblast posunout od refe-
renční buňky.
Sloupce: Tento parametr určuje, o kolik sloupců se má výsledná oblast posunout od refe-
renční buňky.

230 Část II: Ovládnutí grafů

Výška: Tento parametr určuje počet řádků výsledné oblasti.
Šířka: Poslední parametr určuje počet sloupců výsledné oblasti.

Poznámka: Obsahují-li sloupce použité pro data libovolné jiné záznamy, funkce POČET2 vrátí 
nesprávnou hodnotu. Chcete-li si věci ulehčit, nezadávejte do sloupce žádná jiná data. Obsahu-
je-li sloupec další informace, budete muset upravit parametr výška ve funkci POČET2.

Připomeňme si, že pojmenovaný vzorec Prodej byl definován jako:
=POSUN(List1!$B$2;0;0;POČET2(List1!$B:$B)-1;1)

Pokud by se ve sloupci B nacházelo 9 záznamů (v oblasti B1:B9), funkce POČET2 by vrátila 9. 
Tento výsledek musí být snížen o 1, aby bral v úvahu záhlaví sloupce. Proto by pojmenovaný 
vzorec mohl být vyjádřen následovně:
=POSUN(List1!$B$2;0;0;8;1)

Tento vzorec používá buňku B2 jako referenční buňku a vrací odkaz na oblast, která je:
Posunuta od buňky B2 o 0 řádků (druhý parametr – řádky).
Posunuta od buňky B2 o 0 sloupců (třetí parametr – sloupce).
Osm buněk vysoká (čtvrtý parametr – výška).
Jednu buňku široká (pátý parametr – šířka).

Jinými slovy, funkce POSUN vrací odkaz na oblast B2:B9, a to je oblast použitá datovou řadou 
grafu. Přidáte-li nový bod řady, funkce POSUN vrátí odkaz na oblast B2:B10.
Následující příklady této kapitoly používají stejný základní koncept, ale liší se v parametrech 
dodaných do funkce POSUN.

Poznámka: Pro zjednodušení grafy v této kapitole používají pouze jednu datovou řadu. Tyto 
metody však můžete použít i v grafech s mnoha datovými řadami. Budete však muset provést 
nezbytné úpravy u každé řady grafu zvlášť.

Ovládání řady pomocí posuvníku
Příklad v následujících částech představuje jiný typ interaktivity. Obrázek 7.4 znázorňuje graf, 
který používá ovládací prvek posuvník k určení počtu měsíců (od 1 do 12), které se mají v gra-
fu zobrazit.

Příklady v této části jsou dostupné na přiloženém CD. Jméno souboru je graf s posuvníkem.
xlsx.

Vytvoření grafu

Prohlédněte si obrázek 7.4 a použijte data ve sloupcích A:B k vytvoření grafu. (V příkladu je 
použit běžný prostorový sloupcový graf.) Jelikož se počet bodů řady bude lišit, je moudré vypnout 
automatické měřítko svislé osy. Maximum měřítka osy nastavte na dostatečně velkou hodnotu, 
do které se vejdou všechna data (v tomto případě 25 000). Díky tomu osa hodnot zůstane stej-
ná bez ohledu na počet zobrazených bodů řady.

Obrázek 7.4: Ovládací prvek posuvník na vrchu grafu určuje, kolik měsíců se má zobrazit.

Definování názvů 

Tento příklad používá několik názvů, které jsou popsány v této části.
Buňka E1 obsahuje hodnotu, která určuje počet měsíců zobrazených v grafu. Z praktických 
důvodů je tato buňka pojmenovaná PočetMěsíců.
Kromě toho má sešit dva pojmenované vzorce, které jsou použité v datové řadě grafu. 
Vzorec Měsíc vypadá následovně:
=POSUN(List1!$A$2;0;0;PočetMěsíců;1)

Vzorec Zůstatek vypadá následovně:
=POSUN(List1!$B$2;0;0;PočetMěsíců;1)

Chápete-li, jak pojmenované vzorce fungovaly v předchozím příkladu, neměli byste mít problém 
pochopit i tuto variantu. Jak můžete vidět, funkce POSUN používají PočetMěsíců pro parametr 
výšky. Výsledkem je to, že buňka PočetMěsíců ovládá, kolik bodů řady se zobrazuje v grafu.

Tip: Jiný, o něco jednodušší, způsob je definovat Zůstatek jako posun od oblasti Měsíc. Tímto 
způsobem by definice Zůstatku vypadala následovně:

=POSUN(Měsíc;0;1)

Stejně jako u předchozího příkladu, jsou tyto dva pojmenované vzorce následně použity pro 
popisky kategorie a oblast hodnot datové řady grafu. To se provádí pomocí dialogového okna 
Vybrat zdroj dat. Konečný efekt? Změníte-li hodnotu v buňce E1, graf se automaticky aktua-
lizuje.

Přidání ovládacího prvku posuvník

Posuvník zde není vyloženě nutný, ale může být praktický. Posouvání myší je přece jen jedno-
dušší než změna hodnoty v buňce E1.

Poznámka: Předchozí verze Excelu vám umožňovaly přidat ovládací prvek do samotného 
vloženého grafu. To bylo praktické, neboť se ovládací prvek přesouval spolu s grafem. V Excelu 
2007 bohužel ovládací prvek do vloženého grafu vložit nelze. Posuvník tak musíte přidat na list. 
Všimněte si také, že vám Excel nedovolí ani seskupit posuvník s grafem.

Chcete-li do listu přidat ovládací prvek, Excel musí na pásu zobrazovat kartu Vývojář. Pokud ji nevi-
díte, vyberte tlačítko Office  Možnosti aplikace Excel, klepněte na záložku Oblíbené a zaškrtněte 
políčko volby Zobrazit na pásu kartu Vývojář.

Následující postup přidá do listu ovládací prvek posuvník a propojí jej s buňkou E1:
1. Vyberte Vývojář  Ovládací prvky  Vložit a v sekci Ovládací prvky formuláře klepněte 

na ikonu Posuvník.
2. Klepnutím a tažením v listu vytvořte ovládací prvek. Jeho velikost i umístění můžete 

následně upravit jako u libovolného jiného grafického objektu.
3. Klepněte pravým tlačítkem myši na posuvník a vyberte z místní nabídky volbu Formát 

ovládacího prvku, čímž otevřete dialogové okno Formát ovládacího prvku.
4. V dialogu Formát ovládacího prvku klepněte na záložku Ovládací prvek (viz obrázek 

7.5).

Obrázek 7.5: Propojení posuvníku s buňkou.

 5. Zadejte 1 do pole Nejnižší hodnota. Do pole Nejvyšší hodnota zadejte 12 (maximální 
počet bodů řady v grafu).

6. Pole Přírůstková změna nastavte na hodnotu 1 a pole Změna o stránku na hodnotu 3.

7. Do pole Propojení s buňkou zadejte PočetMěsíců. Tímto posuvník propojíte s buňkou 
E1 (pojmenovanou jako PočetMěsíců).

8. Klepnutím na OK zavřete dialog.
Poté, co dokončíte tento postup, hodnota v buňce E1 bude ovládána posuvníkem a bude mít 
rozsah od 1 do 12. Tato hodnota následně určí počet bodů zobrazených v datové řadě grafu.

Poznámka: Excel nabízí dva obecné typy ovládacích prvků: ovládací prvky formuláře a ovládací 
prvky ActiveX. Ovládací prvky formuláře mají jednodušší použití, ale nenabízejí takovou flexibilitu 
jako ovládací prvky ActiveX. Ovládací prvky formuláře např. nenabízejí žádné možnosti formáto-
vání. Pro zjednodušení všechny příklady v této kapitole používají ovládací prvky formuláře.

Určení počátečního a koncového bodu řady
Používá-li graf hodně dat, mohli byste chtít mít možnost omezit množství dat, která se v grafu 
zobrazují. Příklad vidíte na obrázku 7.6.

Obrázek 7.6: Buňky E2 a E4 ovládají množství dat zobrazených v grafu.

Buňka E2 obsahuje hodnotu představující první řádek, který se má zobrazit, a buňka E4 obsahuje 
hodnotu představující poslední řádek, který se má zobrazit. Graf zobrazuje data v řádcích 6 až 
13. Změníte-li hodnoty v buňkách E2 a E4, graf se patřičně přizpůsobí. Tento příklad používá 
ovládací prvky číselník propojené s buňkami E2 a E4. Tyto ovládací prvky zjednodušují úpra-
vy hodnot v těchto buňkách.

Příklady v této části jsou dostupné na přiloženém CD. Jméno souboru je první a poslední 
bod řady.xlsx.

Vytvoření grafu

Prohlédněte si obrázek 7.6 a vytvořte graf z dat ve sloupcích A:C. V příkladu je použit běžný 
sloupcový graf používající sloupce A:B pro popisky osy kategorie, z čehož plynou dva řádky 
popisků kategorie v grafu.

Definování názvů

Buňka E2 má z praktických důvodů název PrvníŘádek a buňka E4 název PosledníŘádek.
Mimo to má sešit dva pojmenované vzorce. Vzorec Datum je definován následovně:
=POSUN(List1!$A$1;PrvníŘádek-1;0;PosledníŘádek-PrvníŘádek+1;2)

Jelikož popisky kategorie zabírají dva sloupce, funkce POSUN jako svůj poslední parametr 
používá 2. Jinými slovy, funkce vrací oblast, která je dva sloupce široká.
Vzorec Prodej je definován následovně:
=POSUN(List1!$C$2;PrvníŘádek-2;0;PosledníŘádek-PrvníŘádek+1;1)

Eventuálně můžete vzorec Prodej definovat pomocí funkce Datum:
=POSUN(Datum;0;2;;1)

Poté, co tyto pojmenované vzorce vytvoříte, jsou pomocí dialogu Vybrat zdroj dat (nebo pří-
mou úpravou vzorce SADA) určeny jako popisky kategorie a oblast hodnot datové řady grafu. 
Další informace o použití pojmenovaných vzorců v datové řadě grafu najdete v části „Úprava 
řady“, výše v této kapitole.

Poznámka: Počáteční verze Excelu 2007 obsahuje závažnou chybu. Zadáte-li nečíselnou hod-
notu do buňky E2 nebo E4, pojmenované vzorce vrátí chybové hodnoty. Excel zobrazí poměrně 
neinformativní chybové hlášení zobrazené na obrázku 7.7 a datová řada z grafu pravděpodobně 
zmizí! Chcete-li, aby se řada opět objevila, opravte chybný záznam a uložte sešit pod jiným názvem 
souboru. Doufejme, že tento problém bude opraven v budoucí aktualizaci.

Obrázek 7.7: Zadání hodnoty způsobující chybu v pojmenovaném vzorci vyvolá chybové hlášení.

Přidání ovládacího prvku číselník

Pro extra komfort můžete do listu přidat ovládací prvek číselník, jehož prostřednictvím bude 
snazší měnit hodnoty PrvníŘádek a PosledníŘádek. Postupujte následovně:

1. Vyberte Vývojář  Ovládací prvky  Vložit a v sekci Ovládací prvky formuláře klepněte 
na ikonu Číselník.

2. Klepnutím a tažením v listu vytvořte ovládací prvek. Jeho velikost i umístění můžete 
následně upravit jako u libovolného jiného grafického objektu.

3. Klepněte pravým tlačítkem myši na číselník a vyberte z místní nabídky volbu Formát 
ovládacího prvku, čímž otevřete dialogové okno Formát ovládacího prvku.

4. V dialogu Formát ovládacího prvku klepněte na záložku Ovládací prvek.
5. Do pole Nejnižší hodnota zadejte 2. 
6. Do pole Nejvyšší hodnota zadejte 25 (nebo jiné číslo, které odpovídá řádku obsahující-

mu poslední bod datové řady).
7. Do pole Propojení s buňkou zadejte PrvníŘádek. Tímto číselník propojíte s buňkou 

E2.
8. Klepnutím na OK zavřete dialog.
9. Zopakujte kroky 3 až 8 pro přidání dalšího číselníku pro buňku PosledníŘádek. V 7. Kroku 

do pole Propojení s buňkou zadejte PosledníŘádek.
Poté, co tento postup dokončíte, budete moci tyto propojené číselníky používat k rychlé změně 
hodnot určujících první a poslední zobrazený bod v grafu.

Určení prvního bodu a počtu bodů řady
Následující příklad je podobný předcházejícímu. Místo toho, abyste uživatele nechali určit první 
a poslední zobrazený řádek, jej však necháte určit první řádek (nějaké smysluplné datum) a počet 
zobrazených bodů řady.
Obrázek 7.8 znázorňuje list obsahující denní prodejní záznamy. Buňka D2 obsahuje první datum, 
které se má zobrazit, a buňka D4 obsahuje počet bodů řady, které se v grafu mají zobrazit.
Tento příklad využívá dvě (volitelná) vylepšení uživatelského prostředí: rozevírací seznam pro 
výběr dne a číselník pro určení počtu dnů. Rozevírací seznam (není na obrázku vidět) byl vytvo-
řen pomocí nástroje Ověření dat.

Příklady v této části jsou dostupné na přiloženém CD. Jméno souboru je první bod a počet 
bodů.xlsx.

Vytvoření grafu

Použijte data v sloupcích A:B k vytvoření grafu. Graf na obrázku je běžný spojnicový, ale tato 
metoda bude fungovat s libovolným typem grafu.

Definování názvů

V tomto příkladu má buňka C2 název PočátečníDen a buňka C4 název PočetDnů. Sešit má dva 
pojmenované vzorce. Vzorec Datum je definován následovně:
=POSUN(List1!$A$2;POZVYHLEDAT(PočátečníDen;List1!$A:$A;1)-2;0;PočetDnů;1)

Vzorec Prodej je definován následovně:
=POSUN(List1!$A$2;POZVYHLEDAT(PočátečníDen;List1!$A:$A;1)-2;1;PočetDnů;1)

Obrázek 7.8: Buňka D2 obsahuje počáteční den a buňka D4 počet dnů, které se mají zobrazit.

Druhý parametr funkce POSUN používá funkci POZVYHLEDAT, která vrací relativní pozi-
ci položky v oblasti. V tomto případě vrátí pozici data dne ve sloupci A, které odpovídá datu 
v buňce PočátečníDen. Toto je samozřejmě jen další způsob, jak určit první řádek, který se má 
v grafu zobrazit.
Stejně jako v předchozím případě jsou pak tyto pojmenované vzorce použity pro popisky katego-
rie a oblast hodnot datové řady grafu. Další informace o použití pojmenovaných vzorců v dato-
vé řadě grafu najdete v části „Úprava řady“, výše v této kapitole.

Přidání prvků uživatelského prostředí

Buňka PočetDnů je propojena s ovládacím prvkem číselník, který ulehčuje určování počtu dnů, 
které se v grafu mají zobrazit (další informace o přidání propojeného číselníku najdete v před-
chozí části).
Pro buňku Počáteční den číselník použít nelze, neboť potřebuje zobrazovat data dnů a číselník 
má maximální hodnotu 30 000. (Pořadová čísla dnů tuto hodnotu překračují.) Jednou možnos-
tí je posuvník, ale rozevírací seznam dostupných dnů bude perfektní. Jeho přidání do buňky je 
pomocí nástroje Ověření dat naštěstí velmi jednoduché. Postupujte následovně:

1. Označte buňku D2 a ujistěte se, že je naformátována, aby zobrazovala datum.
2. Vyberte Data  Datové nástroje  Ověření dat, čímž otevřete dialogové okno Ověření 

dat.
3. V dialogu Ověření dat klepněte na záložku Nastavení. 
4. V poli Povolit vyberte Seznam.

5. Do pole Zdroj zadejte =$A$2:$A$60, což je oblast listu obsahující data dnů (viz obrá-
zek 7.9).

6. Klepnutím na tlačítko OK zavřete dialog Ověření dat.

Obrázek 7.9: Určení oblasti dat dnů pro rozevírací seznam ověření dat.

Poté, co upravíte nastavení ověření dat, můžete klepnout na buňku D2 a vybrat datum z roze-
víracího seznamu. Zvolené datum bude prvním dnem v grafu. Pomocí číselníku určíte, kolik 
bodů řady se v grafu zobrazí.

Tip: V 5. kroku předchozího postupu můžete zvolit jiný přístup. Namísto adresy oblasti můžete do 
pole Zdroj zadat následující vzorec, který se automaticky přizpůsobí, přidáte-li další dny:

=POSUN($A2;0;0;POČET2($A:$A)-1;1)

Zobrazení posledních n bodů datové řady
Další variantou interaktivního grafu je graf zobrazující pouze n posledních bodů řady. Můžete 
například vytvořit graf, který stále zobrazuje pouze data za posledních 6 měsíců (viz obrázek 
7.10). V tomto příkladu buňka F1 obsahuje počet bodů řady, které se v grafu mají zobrazit.
Příklady v této části jsou dostupné na přiloženém CD. Jméno souboru je zobrazení posledních  
n bodů datové řady.xlsx.

Vytvoření grafu

Použijte data v sloupcích A:C k vytvoření grafu. Graf na obrázku je běžný sloupcový, ale tato 
metoda bude fungovat s libovolným typem grafu. Vodorovná osa kategorie používá dva sloup-
ce (A a B).

Obrázek 7.10: Tento graf zobrazuje 6 nejnovějších bodů řady. Jejich počet určuje hodnota v buňce F1.

Definování názvů

V tomto příkladu má buňka F1 název PočetMěsíců. Sešit má dva pojmenované vzorce. Vzorec 
Datum je definován následovně:
=POSUN(List1!$A$2;POČET2(List1!$B:$B)-PočetMěsíců-1;0;PočetMěsíců;2)

Vzorec Prodej je definován následovně:
=POSUN(List1!$C$2;POČET2(List1!$C:$C)-PočetMěsíců-1;0;PočetMěsíců;1)

Název grafu používá odkaz na buňku H1, která obsahuje následující vzorec:
=”Měsíční prodej (za posledních „ &PočetMěsíců&” měsíců)”

Tento vzorec používá název buňky PočetMěsíců, aby zajistil, že se v názvu grafu bude vždy zob-
razovat počet zobrazených měsíců.
Poté, co vytvoříte názvy, použijete tyto dva pojmenované vzorce pro popisky kategorie a oblast 
hodnot datové řady grafu. Další informace o použití pojmenovaných vzorců v datové řadě grafu 
najdete v části „Úprava řady“, výše v této kapitole. Počet bodů řady zobrazených v grafu bude určo-
ván hodnotou v buňce F1. Graf se novým datům přidaným do listu automaticky přizpůsobí.

Zobrazení každého n-tého bodu datové řady
Představte si, že máte v sloupci obrovské množství dat a chcete zobrazit pouze každý desátý 
bod řady. Následující části vám představí dvě metody, které vám to umožní. Tento postup je 
nejvhodnější pro velké sady dat, jejichž data se mění hladce a plynule. V některých případech 
můžete skrýt důležitá data, nezobrazí-li se některé z bodů řady.

Příklady v této části jsou dostupné na přiloženém CD. Jméno souboru je zobrazení každé n-té 
hodnoty.xlsx.

Použití filtrování

Jedním ze způsobů zobrazení každého n-tého bodu oblasti je použití filtrování společně se vzor-
cem. Filtrování vám umožní skrýt řádky, které nevyhovují zadaným parametrům. Excel stan-
dardně nezobrazuje data, která se nacházejí ve skrytém řádku. Z tohoto důvodu je třeba vytvořit 
vzorce, které vrátí určitou hodnotu založenou na čísle řádku dat, a poté výsledky těchto vzorců 
použít jako základ pro filtrování seznamu.
Obrázek 7.11 znázorňuje sešit s aktivním filtrováním (všimněte si skrytých řádků). Buňka F1 
obsahuje hodnotu, která představuje n. Je-li n např. 5, graf zobrazí pouze každý pátý bod řady, 
tj. hodnoty v řádcích 2, 7, 12 atd. 

Poznámka: Máte-li data v tabulce (vytvořené pomocí Vložení  Tabulky  Tabulka), filtrová-
ní je v tabulce automaticky zapnuto. Pokud data v tabulce nemáte, vyberte Data  Seřadit 
a filtrovat  Filtr, čímž filtrování povolíte. V obou případech bude řádek záhlaví v každé buňce 
zobrazovat rozevírací šipky. Chcete-li nastavit filtr sloupce, klepněte na šipku. 

Obrázek 7.11: Graf zobrazuje každý n-tý bod řady (určený v buňce F1) tím, že ignoruje data v řádcích skrytých 
fi ltrováním.

Sloupec A obsahuje 365 dnů a sloupec B obsahuje 365 odpovídajících bodů řady. Sloupec C 
obsahuje vzorce vracející hodnotu, která se použije k zjištění, zda má být řádek skryt. Vzorec 
v buňce C2, který je zkopírován do buněk níže, vypadá následovně:
=MOD(ŘÁDEK()-ŘÁDEK($B$2);$F$1)

Tento vzorec používá funkci MOD k výpočtu zbytku po dělení rozdílu čísla řádku a čísla prv-
ního řádku hodnotou v F1. V důsledku toho každá n-tá buňka sloupce C obsahuje 0.
Pomocí rozevírací šipky v buňce C1 nechejte zobrazit pouze ty řádky, které v sloupci C obsahují 
0. Tato metoda nefunguje, je-li v grafu aktivní volba Zobrazit data ve skrytých řádcích a sloup-
cích. Ta je standardně zakázaná. Chcete-li její nastavení zkontrolovat (nebo změnit), označte 
graf a vyberte Nástroje grafu  Návrh  Vybrat data. V dialogovém okně Vybrat zdroj dat klep-
něte na tlačítko Skryté a prázdné buňky, čímž otevřete dialog, který vám umožní nastavit, jak 
se mají skryté buňky zobrazovat.

Tip: Máte-li graf vedle dat, zjistíte, že se výška grafu sníží, jsou-li skryté řádky. Chcete-li tomu 
zabránit, aktivujte graf a otevřete dialogové okno Velikost a vlastnosti. To provedete klepnutím 
na spouštěč dialogového okna ve skupině Nástroje grafu  Formát  Velikost. (Spouštěč dialogo-
vého okna je ta malá ikona napravo od slova Velikost). V dialogu Velikost a vlastnosti klepněte na 
záložku Vlastnosti a vyberte volbu Přesun buněk bez objektu.

Poznámka: Hlavní problém této metody spočívá v tom, že není plně automatická. Změníte-li hod-
notu v buňce F1, musíte parametry filtru v sloupci C určit znovu. Řádky se automaticky neskryjí.

Použití vzorců pole

Předchozí metoda sice funguje dobře, ale bylo by hezké, kdyby byla plně automatická. Tushar 
Mehta, expert na grafy v Excelu, vymyslel chytrou metodu, která využívá pojmenované vzorce. 
Příklad v této části je adaptací jeho metody.
Obrázek 7.12 znázorňuje stejná data, jež byla použita v předchozím příkladu. Tento sešit použí-
vá tři pojmenované oblasti: Ntý (buňka H1), Datum (oblast A2:A366) a Data (oblast B2:B366). 
Buňka Ntý je propojená s číselníkem.
Všimněte si dalších dvou sloupců (sloupce D a E). Oba z nich jsou vzorci typu pole. Vzorec typu 
pole v D2:D366 vypadá následovně:
=N(POSUN(Datum;(ŘÁDEK(POSUN($A$1;0;0;ŘÁDKY(Datum)/Ntý))-1)*Ntý;0))

Tento vzorec vrací pole skládající se z každého n-tého řádku oblasti Datum.
Následující vzorec typu pole v E2:E366 je velmi podobný a vrací pole skládající se z každého 
n-tého řádku oblasti Data:
=N(POSUN(Data;(ŘÁDEK(POSUN($A$1;0;0;ŘÁDKY(Data)/Ntý))-1)*Ntý;0))

Chcete-li zadat vzorec typu pole do několika buněk, nejprve buňky označte. Poté vzorec zadejte 
do řádku vzorců a stiskněte Ctrl+Shift+Enter.

Obrázek 7.12: Použití pojmenovaných vzorců k zobrazení každého n-tého bodu řady.

Poznámka: Vzorce typu pole jsou složité a jejich kompletní popis by přesahoval rámec této knihy. 
Nemusíte jim však zcela rozumět, abyste je mohli používat. Můžete je jednoduše upravit pro použi-
tí s jinými sadami dat. Buďte si však vědomi, že odkaz na buňku $A$1 musí zůstat netknutý. Tato 
buňka se používá pro generování řady posunů, které odkazují na buňky uvnitř původní oblasti.

Vytvoření pojmenovaných vzorců

Vytvoříte-li graf z dat v sloupcích D:E, výsledek nebude příliš uspokojivý. Zobrazila by se sice 
každá n-tá hodnota, ale graf by zobrazil jen dlouhou řadu prázdných buněk (#N/A).
Řešením je využití metody pojmenovaných vzorců, které budou suplovat oba vzorce typu pole. 
Tyto pojmenované vzorce jsou identické s vzorci typu pole uvedenými v předchozí části.
NovýDatum definujte následovně:
=N(POSUN(Datum;(ŘÁDEK(POSUN(List2!$A$1;0;0;ŘÁDKY(Datum)/Ntý))-1)*Ntý;0))

NováData definujte následovně:
=N(POSUN(Data;(ŘÁDEK(POSUN(List2!$A$1;0;0;ŘÁDKY(Data)/Ntý))-1)*Ntý;0))

Poté, co vytvoříte názvy, použijete tyto dva pojmenované vzorce pro popisky kategorie a oblast 
hodnot datové řady grafu. Další informace o použití pojmenovaných vzorců v datové řadě grafu 
najdete v části „Úprava řady“, výše v této kapitole. A jaký bude výsledek? Pole použitá grafy se 
budou skládat pouze ze skutečných hodnot (žádných #N/A).
Jelikož jsou pojmenované vzorce náhradou za vzorce pole, vzorce v sloupcích D:E už nejsou 
dále potřebné.

Použití zaškrtávacích políček k výběru 

zobrazených řad 
Příklad zobrazený na obrázku 7.13 znázorňuje spojnicový graf se třemi datovými řadami. Počet 
řad, které se skutečně zobrazují, je ovládán třemi ovládacími prvky zaškrtávací políčko. Jsou-li 
zaškrtnuta všechna tři políčka, graf zobrazí data produktu A, produktu B i produktu C. Odstra-
níte-li zaškrtnutí políčka, odpovídající řada z grafu zmizí.

Příklady v této části jsou dostupné na přiloženém CD. Jméno souboru je vybrání řady zaškr-
távacím políčkem.xlsx.

Obrázek 7.13: Zaškrtávací políčka ovládají, které řady se v grafu zobrazují.

Vytvoření grafu 

V tomto příkladu je použit běžný spojnicový graf používající data v A1:D13.

Přidání ovládacího prvku zaškrtávací políčko

Tato část popisuje, jak do listu přidat zaškrtávací políčka a jak každé z nich propojit s buňkou.

Poznámka: Chcete-li do listu přidat ovládací prvek, Excel musí na pásu zobrazovat kartu Vývojář. 
Pokud ji nevidíte, vyberte tlačítko Office  Možnosti aplikace Excel, klepněte na záložku Oblíbené 
a zaškrtněte políčko volby Zobrazit na pásu kartu Vývojář.

Chcete-li do listu přidat zaškrtávací políčko, postupujte následovně:
1. Vyberte Vývojář  Ovládací prvky  Vložit a v sekci Ovládací prvky formuláře klepněte 

na ikonu Zaškrtávací políčko.
2. Klepnutím a tažením v listu vytvořte ovládací prvek. Jeho velikost i umístění můžete 

následně upravit jako u libovolného jiného grafického objektu.
3. Klepněte pravým tlačítkem myši na zaškrtávací políčko a vyberte z místní nabídky volbu 

Formát ovládacího prvku, čímž otevřete dialogové okno Formát ovládacího prvku.
4. V dialogu Formát ovládacího prvku klepněte na záložku Ovládací prvek.
5. Do pole Propojení s buňkou zadejte $G$4. Tímto zaškrtávací políčko propojíte s buň-

kou G4, která v závislosti na stavu zaškrtávacího políčka zobrazí buď PRAVDA, nebo 
NEPRAVDA.

6. Klepnutím na OK zavřete dialog.
7. Opakujte koky 2 až 6 a přidejte další dvě zaškrtávací políčka propojená s buňkami G5 

a G6.

Definování názvů

Tento příklad používá poměrně hodně názvů, které jsou vypsané v tabulce 7.1. Všimněte si, že 
ŘadaA, ŘadaB a ŘadaC jsou pojmenované vzorce. Všechny ostatní názvy odkazují na buňky 
či oblasti. Rovněž si všimněte, že oblast E2:E13 je prázdná. Tato oblast bude použita, nebude-li 
zvolena žádná řada.

Tabulka 7.1: Defi nované názvy

Název Odkaz na

Měsíc =List1!$A$2:$A$13

PrázdnáOblast =List1!$E$2:$E$13

ProduktA =List1!$B$2:$B$13

ProduktB =List1!$C$2:$C$13

ProduktC =List1!$D$2:$D$13

ŘadaA =KDYŽ(UkažProduktA;ProduktA;PrázdnáOblast)

ŘadaB =KDYŽ(UkažProduktB;ProduktB;PrázdnáOblast)

Název Odkaz na

ŘadaC =KDYŽ(UkažProduktC;ProduktC;PrázdnáOblast)

UkažProduktA =List1!$G$4

UkažProduktB =List1!$G$5

UkažProduktC =List1!$G$6

Tyto tři pojmenované vzorce se od předchozích příkladů v kapitole liší. Tyto vzorce používají 
funkci KDYŽ, která kontroluje hodnotu odpovídajícího zaškrtávacího políčka (uloženou v buň-
ce sloupce G). Je-li hodnota PRAVDA, pojmenovaný vzorec vrátí odkaz na pole dat odpovída-
jícího produktu. Pokud políčko není zaškrtnuto, pojmenovaný vzorec vrátí odkaz na prázdnou 
oblast (E2:E13).

Úprava datové řady grafu

Posledním krokem je úprava tří datových řad tak, aby pro oblast hodnot používaly pojmeno-
vané vzorce. Nejjednodušší způsob, jak to udělat, je přímá úprava vzorce SADA. Např. vzorec 
SADA pro Produkt A vypadá následovně:
=SADA(List1!$B$1;List1!Měsíc;List1!ŘadaA;1)

Můžete také použít dialogové okno Vybrat zdroj dat.
Řady Produkt B a Produkt C jsou upraveny podobným způsobem.

Poznámka: Namísto legendy jsem použil popisky dat (jeden pro každou řadu). Legenda totiž 
zobrazuje všechny tři řady – a to i tehdy, když se nějaká řada v grafu vlastně nezobrazuje.

Vytvoření velmi interaktivního grafu
Poslední příklad, zobrazený na obrázku 7.14, je užitečná aplikace, která uživateli umožňuje 
vybrat dvě americká města (ze seznamu 284 měst) a zobrazit graf, který města porovná v jed-
notlivých měsících podle následujících kategorií: průměrné srážky, průměrná teplota, procento 
slunečného počasí a průměrná rychlost větru. 

Příklady v této části jsou dostupné na přiloženém CD. Jméno souboru je graf povětrnostních 
podmínek.xlsx.

Interaktivitu poskytují vestavěné nástroje Excelu, žádná makra nejsou potřeba. Města se vybírají 
z rozevíracího seznamu pomocí nástroje Ověření dat a zobrazovaná data se ovládají pomocí čtyř 
ovládacích prvků přepínače. Všechny části jsou propojené pomocí několika vzorců.

Poznámka: Tento příklad používá několik pojmenovaných oblastí. Ale, na rozdíl od předchozích 
příkladů uvedených v této kapitole, nepoužívá pojmenované vzorce. Místo toho graf používá 
data, která se získají pomocí vzorců SVYHLEDAT.

Tento příklad dokazuje, že je skutečně možné vytvořit uživatelsky přívětivé, interaktivní apli-
kace bez pomoci maker.

Obrázek 7.14: Tato aplikace používá celou řadu metod k zobrazení měsíčních záznamů povětrnostních 
podmínek ve dvou amerických městech.

Následující části popisují postup pro vytvoření této aplikace.

Získání dat

Data jsem našel asi po pěti minutách hledání na Internetu v Národním klimatickém datovém 
středisku (National Climatic Data Center). Data jsem zkopíroval z okna svého webového pro-
hlížeče, vložil do listu Excelu a trochu je pročistil. Výsledkem byly čtyři 13sloupcové tabulky 
dat, které jsem nazval SrážkovýData, TeplotníData, SlunečníData a VětrnýData. Aby uživatelské 
prostředí zůstalo nejjednodušší, dal jsem data na samostatný list (jménem Data).

Vytvoření ovládacího prvku přepínač

Jelikož jsem potřeboval nějaký způsob, jak uživateli umožnit vybrat data, která se mají zobrazit, 
rozhodl jsem se použít ovládací prvek Přepínač z kategorie Ovládací prvky formuláře. Jelikož 
přepínače fungují jako skupina, všechny čtyři přepínače jsou propojené se stejnou buňkou (Q2). 
Buňka Q2 tudíž, v závislosti na vybraném přepínači, obsahuje hodnotu od 1 do 4.
Dále jsem potřeboval způsob, jak na základě číselné hodnoty buňky Q2 získat název tabulky. 
Řešením bylo vytvoření vzorce (v buňce Q3), který využívá funkce ZVOLIT:
=ZVOLIT(Q2;”TeplotníData”;”SrážkovýData”;”SlunečníData”;”VětrnýData”)

Buňka Q3 (nazvaná DataKpoužití) proto zobrazuje název jedné ze čtyř definovaných 
oblastí  dat.

Vytvoření seznamů měst

Dalším krokem v nastavení aplikace je vytvoření rozevíracích seznamů, které uživateli umožní 
vybrat města, která se v grafu budou porovnávat. Díky nástroji Ověření dat je vytvoření rozevíra-
cího seznamu v buňce velmi jednoduchá záležitost. Buňka B4 obsahuje seznam pro první město, 
a proto má název Město1. Buňka B5, nazvaná Město2, obsahuje seznam pro druhé město. 
Aby byla práce se seznamem měst jednodušší, vytvořil jsem definovanou oblast SeznamMěst, 
která odkazuje na první sloupec tabulky SrážkovýData.
Při vytvoření rozevíracích seznamů jsem postupoval následovně:

1. Označte buňku B4.
2. Vyberte Data  Datové nástroje  Ověření dat, čímž otevřete dialogové okno Ověření 

dat.
3. V dialogu Ověření dat klepněte na záložku Nastavení. 
4. V poli Povolit vyberte Seznam.
5. Do pole Zdroj zadejte =SeznamMěst.
6. Klepněte na tlačítko OK.
7. Zkopírujte buňku B4 do B5. Tímto vytvoříte duplikát nastavení ověření dat pro druhé 

město.

Obrázek 7.15: Použití rozevíracího seznamu ověření dat pro výběr města.

Vytvoření oblasti dat grafu

Podstatou této aplikace je, že graf používá data z určité oblasti. Data této oblasti se z příslušné 
oblasti získají použitím vzorců využívajících funkci SVYHLEDAT. Obrázek 7.16 znázorňuje 
oblast dat použitých v grafu.
Vzorec v buňce D4 vyhledávající data v závislosti na obsahu buňky Město1 vypadá následovně :
=SVYHLEDAT(Město1;NEPŘÍMÝ.ODKAZ(DataKpoužití);SLOUPEC()-2;NEPRAVDA)

Předpokládejme například, že Město1 obsahuje Aberdeen, SD a že je zvolen přepínač Srážky. 
Tento přepínač způsobuje, že se v buňce DataKpoužití objeví text SrážkovýData. Vzorec v buň-
ce D4 proto vyhledá řádek v oblasti SrážkovýData, který obsahuje text Aberdeen, SD. Jelikož se 

buňka D4 nachází v sloupci 4, funkce vrátí data z čísla sloupce mínus 2 (tj. ze sloupce 2) oblasti 
SrážkovýData. Tento sloupec obsahuje srážková data z ledna.

Obrázek 7.16: Graf používá data získaná použitím vzorců.

Buňka D4 je zkopírována do ostatních buněk řádku a každý vzorec vrací data svého měsíce.
Vzorec v buňce D5 je stejný s jediným rozdílem – vyhledává data v závislosti na obsahu buňky 
Město2:
=SVYHLEDAT(Město2;NEPŘÍMÝ.ODKAZ(DataKpoužití);SLOUPEC()-2;NEPRAVDA)

Popisek nad měsíci (ve spojených buňkách D2:O2) je generován vzorcem, který odkazuje na 
buňku DataKpoužití a vytváří popisný název. Vzorec vypadá následovně:
=ZLEVA(DataKpoužití;DÉLKA(DataKpoužití)-4)&” průměr”

Vytvoření grafu

Poté, co dokončíte předchozí úkoly, je poslední krok – samotné vytvoření grafu – už procházka 
růžovým sadem. Spojnicový graf má dvě datové řady, používá data v D4:O5 a popisky katego-
rie jsou v D3:O3. Název grafu je propojený s buňkou D2. Data v řádcích 4 a 5 se mění kdykoli 
zvolíte jiný přepínač nebo vyberete jiné město z jednoho ze dvou seznamů ověření dat.

